Documentation

The nb2tex project aims to write a converter that takes Mathematica notebooks and converts them into latex files.

Text cells with tag "tex" become verbatim tex output. They may include anything texish — like formulas $1 + 1 = 2$, or anything else.

Untagged cells are ignored; for example, the following one should not appear in the latex result:

Cells with tag "pdf" are converted into numbered pdf files in a pdf-subfolder (default name: same as the notebook's name; in this case, "Sample"). The latex produced is `\nbpdfType{Sample/nnn.pdf}`, where `Type` is the type of the current cell (`Text`, `Input`, `Output`, `Echo`, etc.), and `nnn` is the number of the current pdf file.

For example:

```
binom = Echo[(a + b)^32] // Expand
```

$$(a + b)^{32}$$

$$a^{32} + 32\,a^{31}\,b + 496\,a^{30}\,b^2 + 4960\,a^{29}\,b^3 + 35\,960\,a^{28}\,b^4 + 201\,376\,a^{27}\,b^5 +$$
$$906\,192\,a^{26}\,b^6 + 3\,365\,856\,a^{25}\,b^7 + 10\,518\,300\,a^{24}\,b^8 + 28\,048\,800\,a^{23}\,b^9 +$$
$$64\,512\,240\,a^{22}\,b^{10} + 129\,024\,480\,a^{21}\,b^{11} + 225\,792\,840\,a^{20}\,b^{12} + 347\,373\,600\,a^{19}\,b^{13} +$$
$$471\,435\,600\,a^{18}\,b^{14} + 565\,722\,720\,a^{17}\,b^{15} + 601\,080\,390\,a^{16}\,b^{16} + 565\,722\,720\,a^{15}\,b^{17} +$$
$$471\,435\,600\,a^{14}\,b^{18} + 347\,373\,600\,a^{13}\,b^{19} + 225\,792\,840\,a^{12}\,b^{20} + 129\,024\,480\,a^{11}\,b^{21} +$$
$$64\,512\,240\,a^{10}\,b^{22} + 28\,048\,800\,a^{9}\,b^{23} + 10\,518\,300\,a^{8}\,b^{24} + 3\,365\,856\,a^{7}\,b^{25} +$$
$$906\,192\,a^{6}\,b^{26} + 201\,376\,a^{5}\,b^{27} + 35\,960\,a^{4}\,b^{28} + 4960\,a^{3}\,b^{29} + 496\,a^{2}\,b^{30} + 32\,a\,b^{31} + b^{32}$$

If a tag is of the form "pdf:Name", a pdf file by the name "NameType.pdf" is created in the pdf folder (where Type is the type of the cell), yet nothing is added to the tex file. This allows to treat certain cells is a special manner. For example, using latex commands that will inline them or use them as side figure. As an example, if the Mathematica input is `20!`, the output is `2 432 902 008 176 640 000`.

If a tag is of the form "pdfXXX", with XXX a string not beginning with the character ":", the string XXX gets added to the `\nbpdf` command, so it becomes `\nbpdfXXXType`. This is useful for graphics inclusions, for example, where a useful tag would be "pdfgraph":

```
plot = Plot3D[Cos[x^2 + y^2], {x, -3, 3}, {y, -3, 3}]
```



Invoking nb2tex (suffixes are automatically added and should not be included):

```
nb2tex[nb_String, opts___Rule];
nb2tex[nb_String, tex_String, opts___Rule];
```

Valid options include:

- `"PDFFolder" -> foldername` (a string).
- `"PDFWidth" -> width` (in inches).

Input cells with tag “exec” get executed using `ToExpression` at the time of their processing (even if they do not have the property “Evaluatable), with no output produced. This is useful for setting / re-setting options within the notebook itself. For example, an input cell with tag exec and content `nb2tex$PDFWidth=10` will allow very wide outputs:

```
binom
```

$$a^{32} + 32\, a^{31}\, b + 496\, a^{30}\, b^2 + 4960\, a^{29}\, b^3 + 35\,960\, a^{28}\, b^4 + 201\,376\, a^{27}\, b^5 + 906\,192\, a^{26}\, b^6 + 3\,365\,856\, a^{25}\, b^7 + 10\,518\,300\, a^{24}\, b^8$$
$$28\,048\,800\, a^{23}\, b^9 + 64\,512\,240\, a^{22}\, b^{10} + 129\,024\,480\, a^{21}\, b^{11} + 225\,792\,840\, a^{20}\, b^{12} + 347\,373\,600\, a^{19}\, b^{13} + 471\,435\,600\, a^{18}\, b^{14}$$
$$601\,080\,390\, a^{16}\, b^{16} + 565\,722\,720\, a^{15}\, b^{17} + 471\,435\,600\, a^{14}\, b^{18} + 347\,373\,600\, a^{13}\, b^{19} + 225\,792\,840\, a^{12}\, b^{20} + 129\,024\,480\, a^{11}$$
$$28\,048\,800\, a^{9}\, b^{23} + 10\,518\,300\, a^{8}\, b^{24} + 3\,365\,856\, a^{7}\, b^{25} + 906\,192\, a^{6}\, b^{26} + 201\,376\, a^{5}\, b^{27} + 35\,960\, a^{4}\, b^{28} + 4960\, a^{3}\, b^{29} + 496$$

Similarly, changing `nb2tex$TeXFileName` will change the file where the latex output is written.

Implementation

```
SetOptions[$FrontEndSession, PrintingStyleEnvironment → "Working"];
nb2tex[nb_String, opts___Rule] := nb2tex[nb, nb, opts];
```

```
nb2tex[nb_String, tex_String, opts___Rule] := Module[
  {notebook, PDFCounter = 0, type, tag, pdfname, cells, cell, c, cl, texfiles = {},
   TeXOut,
   PDFFolder = PDFFolder /. {opts} /. PDFFolder → nb
  },
  nb2tex$TeXFileName = tex <> ".tex";
  nb2tex$PDFWidth = PDFWidth /. {opts} /. PDFWidth → 6.5;
  TeXOut[s_String] := (texfiles = texfiles ⋃ {nb2tex$TeXFileName};
    WriteString[nb2tex$TeXFileName, s]);
  notebook = NotebookGet[NotebookOpen@FileNameJoin[{Directory[], nb <> ".nb"}]];
  If[FileType[PDFFolder] === None, CreateDirectory[PDFFolder]];
  DeleteFile /@ FileNames["*.pdf", PDFFolder];
  cells = Cases[notebook, c_Cell /; Length[c] ≥ 2, ∞];
  Do[
   type = cell[[2]];
   tag = CellTags /. Cases[cell, _Rule] /. CellTags → "";
   Which[
    type == "Text" ∧ tag == "tex", TeXOut[
     StringReplace[cell[[1]], {"’" → "'", "”" → "\""}] <> "\n\n"],
    StringMatchQ[tag, "pdf:" ~~ ___], (
     pdfname = PDFFolder <> "/" <> StringDrop[tag, 4] <> type <> ".pdf";
     Export[pdfname, Join[cell, Cell[PageWidth → 80 nb2tex$PDFWidth / 0.75]]];
     cl = "c:\\drorbn\\bin\\cpdf.exe -scale-page \"0.75 0.75\" " <> pdfname <>
       " -o " <> pdfname;
     Close@OpenRead["!" <> cl];
    ),
    StringMatchQ[tag, "pdf" ~~ ___], (
     pdfname = PDFFolder <> "/" <> ToString[++PDFCounter] <> ".pdf";
     Export[pdfname, Join[cell, Cell[PageWidth → 80 nb2tex$PDFWidth / 0.75]]];
     cl = "c:\\drorbn\\bin\\cpdf.exe -scale-page \"0.75 0.75\" " <> pdfname <>
       " -o " <> pdfname;
     Close@OpenRead["!" <> cl];
     TeXOut[StringReplace[
       "\\nbpdfXXXType{pdfname}\n",
       {"XXX" → StringDrop[tag, 3], "Type" → type, "pdfname" → pdfname}
      ]]
    ),
    type == "Input" ∧ tag == "exec", ToExpression[cell[[1]]],
    True, Null
   ],
   {cell, cells}
  ];
  Close /@ texfiles;
 ]
```

Run

```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\nb2tex"];
nb2tex["nb2tex", PDFFolder → "Snips"];
Run@
 "\"C:\\Users\\drorb\\AppData\\Local\\Programs\\MiKTeX\\miktex\\bin\\x64\\pdflatex.exe
   \" nb2tex.tex"
0
```